
ABSTRACT
A picture is worth 1000 words it’s said.  And while that may be true,
producing a picture instead of words can sometimes be 1000 times
more difficult, particularly if you’re not familiar with SAS/Graph and
Annotate.  And while the Internet is a great medium for graphics,
producing them can be a pain.  Anything that would make the task
more simple would be welcomed.

By using the WIDTH= and HEIGHT= options on an HTML image
tag, you can use a one-pixel GIF file to create an attractive bar chart
without any SAS/Graph or Annotate.  This paper describe how.

INTRODUCTION
In developing a web application displaying Injury Mortality Statistics
for the Centers for Disease Control, we wanted to display the
breakdown of certain injury categories in a meaningful and
understandable way.  Initially we had a report layout which reported
numbers and percentages but seemed to lack a certain amount of
impact (see Appendix 1).

The report tells the user everything he wants, but doesn’t necessarily
make the statement that it could.  We realized that a graph of the
data would convey the information much more meaningfully (see
Appendix 2). The question was how to do it.

SAS/GRAPH AND ANNOTATE
A SUGI 24 paper from LeRoy Besseler entitled Show Them What’s
Important: Solutions for a Finite Workday in an Era of Information
Overload contained a graph in the format I wanted and the code to
create such a graph using SAS/Graph and Annotate.  It was a good
solution but because some of the labels were so long they wouldn’t
display properly without a lot of extra work.  I wanted an easier
solution.

ANOTHER APPROACH USING HTML
It turned out that using the WIDTH= and HEIGHT= parameters of
the image tag in HTML would generate a bar chart.  All I needed was
to produce a one-pixel .GIF file for the color I wanted, and then
specify the width and height that I wanted it displayed.  I could
generate the HTML dynamically and thus vary the chart based on
the input criteria I needed.  In it’s simplest form, the HTML code
would look like this:

<IMG border=0 height=20 src="red.gif"
width="74.76%">

The HEIGHT parameter would be static since I was doing horizontal
bar charts and the WIDTH parameter (the percentage) I could get
from the output of a PROC FREQ.  (I assume you could reverse the
values of HEIGHT and WIDTH for vertical bar charts but I’ve not
tried it).

Simple, yet it can produce good-looking graphics.  For an example of
what it can look like,  Appendix 2 contains a partial report from our
web application.   The code used to produce this graphic can be
found in Appendix 3.

BENEFITS
There were several advantages to this approach:

1. The graph could be put in a multi-column table which would let the

bar be in one column and the labels be in another.  This would allow
the HTML table to control how the text of the label wrapped and
ensured that it stayed with the horizontal bar.  The application didn’t
have to control that, the browser did.
2. It didn’t require any knowledge of SAS/Graph or Annotate.

3. The resulting HTML file was small which would make downloads
quicker.  Instead of downloading a big graphic to your browser you
end up downloading only a one-pixel .GIF file which speeds
response time.

CONCLUSION
If you don’t know SAS/GRAPH or Annotate very well, you need not
be prevented from putting graphics in your web application.

APPENDIX 1 – NON-GRAPHICAL DISPLAY OF INFORMATION

1997 United States
Homicide and Legal Intervention

Ages 25-34, All Races, Both Sexes
Total Deaths: 5,075

Cause of Death Number of Deaths Percent

Firearm 3,794 74.76
Cut/piercing 575 11.3
Etc. __________ _______

5,075 100%

APPENDIX 2 – GRAPHICAL DISPLAY OF INFORMATION



APPENDIX 3 – SAS CODE

*------------------------------------------------------------------------;
* Sum up death data so that you have the cause, the total number of      ;
* deaths for each cause, and the percentage that total is from all deaths;
*------------------------------------------------------------------------;

proc freq data=temp ;
  tables cause / noprint out=totals(rename=count=deaths) ;
  weight deaths ;
  label cause=’Cause of Injury’ deaths=’Number of Deaths’ ;
  run ;

*------------------------------------------------------------------------;
* Now produce the chart as a table with three columns: one with the cause;
* of death, one with the number of deaths, and one with the percentage of;
* deaths including the bar chart.                                        ;
* The bar chart is drawn using a .GIF that is one pixel in size, and uses;
* the WIDTH= and HEIGHT= options on the IMG tag to make it the right     ;
* size.  In order to fit it to scale, we divide the number of deaths by  ;
* a macro variable so that it fits on one page.                          ;
*------------------------------------------------------------------------;

data _null_ ;
  file _webout  ;
  set work.totals end=lastob;
  if _n_ = 1
  then do ;
       put ’<table border="0" cellpadding="2" width="771">’ //
            ’<tr>’ /
           ’<td width="132" height="27"><font face="Arial">’
           ’<b>Cause of Death</b></font></td>’
           ’<td nowrap><p align="right"><font size="2" face="Arial">’
           ’<b>Number<br> of Deaths</b></font></td>’
           ’<td width="18" height="27"></td>’

           ’<td width="536" height="27">’
           ’<font face="Arial" size="2">’
           ’<b>Percentage of Deaths</b></font></td>’
           ’</tr>’
            ;
        end ;
   * create quoted strings for height and width.  Want them;
   * to look like height="20" and width="44.5%" ;
   length height width $8 total 5 ;
   if percent < 1.0 then width=quote("1.0%") ;  * make small numbers a minimum size ;
   else width = compress(quote(put(percent,5.2) !! ’%’)) ;
   height = quote("20") ;
   total = &deaths ;
   put /* print out the cause of death */
       ’<tr>’
       ’<td width="132" height="30"><font face="Arial">’
       cause
       ’</font></td>’

       /* print out the number of deaths */
       ’<td width="61" height="30" align="right">’
       ’<font face="Arial">’
       deaths comma10.
       ’</font></td>’

        /* put spaces in column to hold it open */
       ’<td width="10" align="right"><font color="#FFFFFF" size="2" ’
       ’face="Arial">’
       ’<img src="http://www.cdc.gov/navimages/spacer.gif" width="5">’
       ’</font></td>’

       /* print out the percentage of deaths and the bar */
       ’<td width="538" height="30" align="left">’



       ’<font face="Arial">’
       ’<img border="0" ’
       ’src="http://www.cdc.gov/ncipc/images/red.gif"’
       ’width=’ width ’height=’ height ’>’
       ’<img border="0" ’
       ’src="http://www.cdc.gov/navimages/spacer.gif" ’
       ’width="5">’
       percent 4.1 ’%</font></td>’
       ’</tr>’
       ;
     if lastob
     then do ;

         put ’<tr>’
             ’<td width="132" height="30"><font face="Arial" size="2">’
             ’<b>Total Deaths&nbsp;</b></font></td>’
             ’<td width="85" height="30" colspan="2"><font size="2">’
             ’<p align="right"></font><font size="2" face="Arial"><b>’
             total comma10.
             ’</b></font><font face="Arial"> </font></td>’
             ’<td width="536" height="30"></td>’
             ’</tr>’
             ;

          put ’</table>’ ;
          end ;
     run ;


